

# NAG Fortran Library Routine Document

## D02RAF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

### 1 Purpose

D02RAF solves the two-point boundary-value problem with general boundary conditions for a system of ordinary differential equations, using a deferred correction technique and Newton iteration.

### 2 Specification

```

SUBROUTINE D02RAF(N, MNP, NP, NUMBEG, NUMMIX, TOL, INIT, X, Y, IY, ABT,
1          FCN, G, IJAC, JACOBF, JACOBG, DELEPS, JACEPS, JACGEP,
2          WORK, LWORK, IWORK, LIWORK, IFAIL)
  INTEGER      N, MNP, NP, NUMBEG, NUMMIX, INIT, IY, IJAC, LWORK,
1          IWORK(LIWORK), LIWORK, IFAIL
  real        TOL, X(MNP), Y(IY,MNP), ABT(N), DELEPS, WORK(LWORK)
  EXTERNAL    FCN, G, JACOBF, JACOBG, JACEPS, JACGEP

```

### 3 Description

D02RAF solves a two-point boundary-value problem for a system of  $n$  ordinary differential equations in the interval  $(a, b)$  with  $b > a$ . The system is written in the form

$$y'_i = f_i(x, y_1, y_2, \dots, y_n), \quad i = 1, 2, \dots, n \quad (1)$$

and the derivatives  $f_i$  are evaluated by a subroutine FCN supplied by the user. With the differential equations (1) must be given a system of  $n$  (nonlinear) boundary conditions

$$g_i(y(a), y(b)) = 0, \quad i = 1, 2, \dots, n,$$

where

$$y(x) = [y_1(x), y_2(x), \dots, y_n(x)]^T. \quad (2)$$

The functions  $g_i$  are evaluated by a subroutine G supplied by the user. The solution is computed using a finite-difference technique with deferred correction allied to a Newton iteration to solve the finite-difference equations. The technique used is described fully in Pereyra (1979).

The user must supply an absolute error tolerance and may also supply an initial mesh for the finite-difference equations and an initial approximate solution (alternatively a default mesh and approximation are used). The approximate solution is corrected using Newton iteration and deferred correction. Then, additional points are added to the mesh and the solution is recomputed with the aim of making the error everywhere less than the user's tolerance and of approximately equidistributing the error on the final mesh. The solution is returned on this final mesh.

If the solution is required at a few specific points then these should be included in the initial mesh. If, on the other hand, the solution is required at several specific points then the user should use the interpolation routines provided in Chapter E01 if these points do not themselves form a convenient mesh.

The Newton iteration requires Jacobian matrices

$$\left(\frac{\partial f_i}{\partial y_j}\right), \left(\frac{\partial g_i}{\partial y_j(a)}\right) \text{ and } \left(\frac{\partial g_i}{\partial y_j(b)}\right).$$

These may be supplied by the user through subroutines JACOBF for  $\left(\frac{\partial f_i}{\partial y_j}\right)$  and JACOBG for the others. Alternatively the Jacobians may be calculated by numerical differentiation using the algorithm described in Curtis *et al.* (1974).

For problems of the type (1) and (2) for which it is difficult to determine an initial approximation from which the Newton iteration will converge, a continuation facility is provided. The user must set up a family of problems

$$y' = f(x, y, \epsilon), \quad g(y(a), y(b), \epsilon) = 0, \quad (3)$$

where  $f = [f_1, f_2, \dots, f_n]^T$  etc., and where  $\epsilon$  is a continuation parameter. The choice  $\epsilon = 0$  must give a problem (3) which is easy to solve and  $\epsilon = 1$  must define the problem whose solution is actually required. The routine solves a sequence of problems with  $\epsilon$ -values

$$0 = \epsilon_1 < \epsilon_2 < \dots < \epsilon_p = 1. \quad (4)$$

The number  $p$  and the values  $\epsilon_i$  are chosen by the routine so that each problem can be solved using the solution of its predecessor as a starting approximation. Jacobians  $\frac{\partial f}{\partial \epsilon}$  and  $\frac{\partial g}{\partial \epsilon}$  are required and they may be supplied by the user via routines JACEPS and JACGEP respectively or may be computed by numerical differentiation.

## 4 References

Pereyra V (1979) PASVA3: An adaptive finite-difference Fortran program for first order nonlinear, ordinary boundary problems *Codes for Boundary Value Problems in Ordinary Differential Equations. Lecture Notes in Computer Science* (ed B Childs, M Scott, J W Daniel, E Denman and P Nelson) **76** Springer-Verlag

Curtis A R, Powell M J D and Reid J K (1974) On the estimation of sparse Jacobian matrices *J. Inst. Maths. Applics.* **13** 117–119

## 5 Parameters

- 1: N – INTEGER *Input*  
*On entry:* the number of differential equations,  $n$ .  
*Constraint:*  $N > 0$ .
- 2: MNP – INTEGER *Input*  
*On entry:* MNP must be set to the maximum permitted number of points in the finite-difference mesh. If LWORK or LIWORK (see below) is too small then internally MNP will be replaced by the maximum permitted by these values. (A warning message will be output if on entry IFAIL is set to obtain monitoring information.)  
*Constraint:*  $MNP \geq 32$ .
- 3: NP – INTEGER *Input/Output*  
*On entry:* NP must be set to the number of points to be used in the initial mesh.  
*Constraint:*  $4 \leq NP \leq MNP$ .  
*On exit:* the number of points in the final mesh.
- 4: NUMBEG – INTEGER *Input*  
*On entry:* the number of left-hand boundary conditions (that is the number involving  $y(a)$  only).  
*Constraint:*  $0 \leq \text{NUMBEG} < N$ .
- 5: NUMMIX – INTEGER *Input*  
*On entry:* the number of coupled boundary conditions (that is the number involving both  $y(a)$  and  $y(b)$ ).  
*Constraint:*  $0 \leq \text{NUMMIX} \leq N - \text{NUMBEG}$ .

- 6: TOL – *real* *Input*  
*On entry:* a positive absolute error tolerance. If
$$a = x_1 < x_2 < \dots < x_{NP} = b$$
is the final mesh,  $z_j(x_i)$  is the  $j$ th component of the approximate solution at  $x_i$ , and  $y_j(x)$  is the  $j$ th component of the true solution of (1) and (2), then, except in extreme circumstances, it is expected that
$$|z_j(x_i) - y_j(x_i)| \leq \text{TOL}, \quad i = 1, 2, \dots, NP; j = 1, 2, \dots, n. \quad (5)$$
*Constraint:* TOL > 0.0.
- 7: INIT – INTEGER *Input*  
*On entry:* indicates whether the user wishes to supply an initial mesh and approximate solution (INIT  $\neq$  0) or whether default values are to be used, (INIT = 0).
- 8: X(MNP) – *real* array *Input/Output*  
*On entry:* the user must set X(1) =  $a$  and X(NP) =  $b$ . If INIT = 0 on entry a default equispaced mesh will be used, otherwise the user must specify a mesh by setting X( $i$ ) =  $x_i$ , for  $i = 2, 3, \dots, NP - 1$ .  
*Constraints:*

$$\begin{aligned} X(1) < X(NP) & \text{ if INIT} = 0, \\ X(1) < X(2) < \dots < X(NP) & \text{ if INIT} \neq 0. \end{aligned}$$
*On exit:* X(1), X(2), ..., X(NP) define the final mesh (with the returned value of NP) and X(1) =  $a$  and X(NP) =  $b$ .
- 9: Y(IY,MNP) – *real* array *Input/Output*  
*On entry:* if INIT = 0, then Y need not be set.  
If INIT  $\neq$  0, then the array Y must contain an initial approximation to the solution such that Y( $j, i$ ) contains an approximation to
$$y_j(x_i), \quad i = 1, 2, \dots, NP; j = 1, 2, \dots, n.$$
*On exit:* the approximate solution  $z_j(x_i)$  satisfying (5) on the final mesh, that is
$$Y(j, i) = z_j(x_i), \quad i = 1, 2, \dots, NP; j = 1, 2, \dots, n,$$
where NP is the number of points in the final mesh. If an error has occurred then Y contains the latest approximation to the solution. The remaining columns of Y are not used.
- 10: IY – INTEGER *Input*  
*On entry:* the first dimension of the array Y as declared in the (sub)program from which D02RAF is called.  
*Constraint:* IY  $\geq$  N.
- 11: ABT(N) – *real* array *Output*  
*On exit:* ABT( $i$ ), for  $i = 1, 2, \dots, n$ , holds the largest estimated error (in magnitude) of the  $i$ th component of the solution over all mesh points.
- 12: FCN – SUBROUTINE, supplied by the user. *External Procedure*  
FCN must evaluate the functions  $f_i$  (i.e., the derivatives  $y'_i$ ) at a general point  $x$  for a given value of  $\epsilon$ , the continuation parameter (see Section 3).

Its specification is:

	SUBROUTINE FCN(X, EPS, Y, F, N)	
	INTEGER	N
	<b>real</b>	X, EPS, Y(N), F(N)
1:	X – <b>real</b>	<i>Input</i>
	<i>On entry:</i> the value of the argument $x$ .	
2:	EPS – <b>real</b>	<i>Input</i>
	<i>On entry:</i> the value of the continuation parameter, $\epsilon$ . This is 1 if continuation is not being used.	
3:	Y(N) – <b>real</b> array	<i>Input</i>
	<i>On entry:</i> the value of the argument $y_i$ , for $i = 1, 2, \dots, n$ .	
4:	F(N) – <b>real</b> array	<i>Output</i>
	<i>On exit:</i> the values of $f_i$ , for $i = 1, 2, \dots, n$ .	
5:	N – INTEGER	<i>Input</i>
	<i>On entry:</i> the number of equations.	

FCN must be declared as EXTERNAL in the (sub)program from which D02RAF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

13: G – SUBROUTINE, supplied by the user. *External Procedure*

G must evaluate the boundary conditions in equation (3) and place them in the array BC.

Its specification is:

	SUBROUTINE G(EPS, YA, YB, BC, N)	
	INTEGER	N
	<b>real</b>	EPS, YA(N), YB(N), BC(N)
1:	EPS – <b>real</b>	<i>Input</i>
	<i>On entry:</i> the value of the continuation parameter, $\epsilon$ . This is 1 if continuation is not being used.	
2:	YA(N) – <b>real</b> array	<i>Input</i>
	<i>On entry:</i> the value $y_i(a)$ , for $i = 1, 2, \dots, n$ .	
3:	YB(N) – <b>real</b> array	<i>Input</i>
	<i>On entry:</i> the value $y_i(b)$ , for $i = 1, 2, \dots, n$ .	
4:	BC(N) – <b>real</b> array	<i>Output</i>
	<i>On exit:</i> the values $g_i(y(a), y(b), \epsilon)$ , for $i = 1, 2, \dots, n$ . These must be ordered as follows:	
	(i) first, the conditions involving only $y(a)$ (see NUMBEG description above);	
	(ii) next, the NUMMIX coupled conditions involving both $y(a)$ and $y(b)$ (see NUMMIX description above); and,	
	(iii) finally, the conditions involving only $y(b)$ ( $N - \text{NUMBEG} - \text{NUMMIX}$ ).	

5:	N – INTEGER	<i>Input</i>
	<i>On entry:</i> the number of equations, $n$ .	

G must be declared as EXTERNAL in the (sub)program from which D02RAF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

14: IJAC – INTEGER *Input*

*On entry:* indicates whether or not the user is supplying Jacobian evaluation routines. If IJAC  $\neq 0$  then the user must supply routines JACOBF and JACOBG and also, when continuation is used, routines JACEPS and JACGEP. If IJAC = 0 numerical differentiation is used to calculate the Jacobian and the routines D02GAZ, D02GAY, D02GAZ and D02GAX respectively may be used as the dummy parameters.

15: JACOBF – SUBROUTINE, supplied by the user. *External Procedure*

JACOBF must evaluate the Jacobian  $\left(\frac{\partial f_i}{\partial y_j}\right)$  for  $i, j = 1, 2, \dots, n$ , given  $x$  and  $y_j$ , for  $j = 1, 2, \dots, n$ .

Its specification is:

SUBROUTINE JACOBF (X, EPS, Y, F, N)		
	INTEGER	N
	<b>real</b>	X, EPS, Y(N), F(N,N)
1:	X – <b>real</b>	<i>Input</i>
	<i>On entry:</i> the value of the argument $x$ .	
2:	EPS – <b>real</b>	<i>Input</i>
	<i>On entry:</i> the value of the continuation parameter $\epsilon$ . This is 1 if continuation is not being used.	
3:	Y(N) – <b>real</b> array	<i>Input</i>
	<i>On entry:</i> the value of the argument $y_i$ , for $i = 1, 2, \dots, n$ .	
4:	F(N,N) – <b>real</b> array	<i>Output</i>
	<i>On exit:</i> F( $i, j$ ) must be set to the value of $\frac{\partial f_i}{\partial y_j}$ , evaluated at the point $(x, y)$ , for $i, j = 1, 2, \dots, n$ .	
5:	N – INTEGER	<i>Input</i>
	<i>On entry:</i> the number of equations, $n$ .	

JACOBF must be declared as EXTERNAL in the (sub)program from which D02RAF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

16: JACOBG – SUBROUTINE, supplied by the user. *External Procedure*

JACOBG must evaluate the Jacobians  $\left(\frac{\partial g_i}{\partial y_j(a)}\right)$  and  $\left(\frac{\partial g_i}{\partial y_j(b)}\right)$ . The ordering of the rows of AJ and BJ must correspond to the ordering of the boundary conditions described in the specification of subroutine G above.

Its specification is:

SUBROUTINE JACOBG(EPS, YA, YB, AJ, BJ, N)		
INTEGER N		
<b>real</b> EPS, YA(N), YB(N), AJ(N,N), BJ(N,N)		
1:	EPS – <b>real</b>	<i>Input</i>
	<i>On entry:</i> the value of the continuation parameter, $\epsilon$ . This is 1 if continuation is not being used.	
2:	YA(N) – <b>real</b> array	<i>Input</i>
	<i>On entry:</i> the value $y_i(a)$ , for $i = 1, 2, \dots, n$ .	
3:	YB(N) – <b>real</b> array	<i>Input</i>
	<i>On entry:</i> the value $y_i(b)$ , for $i = 1, 2, \dots, n$ .	
4:	AJ(N,N) – <b>real</b> array	<i>Output</i>
	<i>On exit:</i> AJ( $i, j$ ) must be set to the value $\frac{\partial g_i}{\partial y_j(a)}$ , for $i, j = 1, 2, \dots, n$ .	
5:	BJ(N,N) – <b>real</b> array	<i>Output</i>
	<i>On exit:</i> BJ( $i, j$ ) must be set to the value $\frac{\partial g_i}{\partial y_j(b)}$ , for $i, j = 1, 2, \dots, n$ .	
6:	N – INTEGER	<i>Input</i>
	<i>On entry:</i> the number of equations, $n$ .	

JACOBG must be declared as EXTERNAL in the (sub)program from which D02RAF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

17: DELEPS – **real** *Input/Output*

*On entry:* DELEPS must be given a value which specifies whether continuation is required. If  $\text{DELEPS} \leq 0.0$  or  $\text{DELEPS} \geq 1.0$  then it is assumed that continuation is not required. If  $0.0 < \text{DELEPS} < 1.0$  then it is assumed that continuation is required unless  $\text{DELEPS} < \sqrt{\text{machine precision}}$  when an error exit is taken. DELEPS is used as the increment  $\epsilon_2 - \epsilon_1$  (see (4)) and the choice  $\text{DELEPS} = 0.1$  is recommended.

*On exit:* an overestimate of the increment  $\epsilon_p - \epsilon_{p-1}$  (in fact the value of the increment which would have been tried if the restriction  $\epsilon_p = 1$  had not been imposed). If continuation was not requested then  $\text{DELEPS} = 0.0$ .

If continuation is not requested then the parameters JACEPS and JACGEP may be replaced by dummy actual parameters in the call to D02RAF. (D02GAZ and D02GAX respectively may be used as the dummy parameters.)

18: JACEPS – SUBROUTINE, supplied by the user. *External Procedure*

JACEPS must evaluate the derivative  $\frac{\partial f_i}{\partial c}$  given  $x$  and  $y$  if continuation is being used.

Its specification is:

SUBROUTINE JACEPS(X, EPS, Y, F, N)		
INTEGER N		
<b>real</b> X, EPS, Y(N), F(N)		
1:	X – <b>real</b>	<i>Input</i>
	<i>On entry:</i> the value of the argument $x$ .	

2:	EPS – <i>real</i> <i>On entry:</i> the value of the continuation parameter, $\epsilon$ .	<i>Input</i>
3:	Y(N) – <i>real</i> array <i>On entry:</i> the solution values $y_i$ at the point $x$ , for $i = 1, 2, \dots, n$ .	<i>Input</i>
4:	F(N) – <i>real</i> array <i>On exit:</i> F( $i$ ) must contain the value $\frac{\partial f_i}{\partial \epsilon}$ at the point $(x, y)$ , for $i = 1, 2, \dots, n$ .	<i>Output</i>
5:	N – INTEGER <i>On entry:</i> the number of equations, $n$ .	<i>Input</i>

JACEPS must be declared as EXTERNAL in the (sub)program from which D02RAF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

19: JACGEP – SUBROUTINE, supplied by the user. *External Procedure*

JACGEP must evaluate the derivatives  $\frac{\partial g_i}{\partial \epsilon}$  if continuation is being used.

Its specification is:

<pre> SUBROUTINE JACGEP (EPS, YA, YB, BCEP, N) INTEGER          N <i>real</i>           EPS, YA(N), YB(N), BCEP(N) </pre>		
1:	EPS – <i>real</i> <i>On entry:</i> the value of the continuation parameter, $\epsilon$ .	<i>Input</i>
2:	YA(N) – <i>real</i> array <i>On entry:</i> the value of $y_i(a)$ , for $i = 1, 2, \dots, n$ .	<i>Input</i>
3:	YB(N) – <i>real</i> array <i>On entry:</i> the value of $y_i(b)$ , for $i = 1, 2, \dots, n$ .	<i>Input</i>
4:	BCEP(N) – <i>real</i> array <i>On exit:</i> BCEP( $i$ ) must contain the value of $\frac{\partial g_i}{\partial \epsilon}$ , for $i = 1, 2, \dots, n$ .	<i>Output</i>
5:	N – INTEGER <i>On entry:</i> the number of equations, $n$ .	<i>Input</i>

JACGEP must be declared as EXTERNAL in the (sub)program from which D02RAF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

20: WORK(LWORK) – *real* array *Workspace*  
21: LWORK – INTEGER *Input*

*On entry:* the dimension of the array WORK as declared in the (sub)program from which D02RAF is called.

*Constraint:*  $LWORK \geq MNP \times (3N^2 + 6N + 2) + 4N^2 + 3N$ .

- 22: IWORK(LIWORK) – INTEGER array *Workspace*  
 23: LIWORK – INTEGER *Input*

*On entry:* the dimension of the array IWORK as declared in the (sub)program from which D02RAF is called.

*Constraints:*

$$\begin{aligned} \text{IWORK} &\geq \text{MNP} \times (2 \times \text{N} + 1) + \text{N} \text{ if } \text{IJAC} \neq 0, \\ \text{IWORK} &\geq \text{MNP} \times (2 \times \text{N} + 1) + \text{N}^2 + 4 \times \text{N} + 2 \text{ if } \text{IJAC} = 0. \end{aligned}$$

- 24: IFAIL – INTEGER *Input/Output*

For this routine, the normal use of IFAIL is extended to control the printing of error and warning messages as well as specifying hard or soft failure (see Chapter P01).

Before entry, IFAIL must be set to a value with the decimal expansion  $cba$ , where each of the decimal digits  $c$ ,  $b$  and  $a$  must have a value of 0 or 1.

$a = 0$  specifies hard failure, otherwise soft failure;

$b = 0$  suppresses error messages, otherwise error messages will be printed (see Section 6);

$c = 0$  suppresses warning messages, otherwise warning messages will be printed (see Section 6).

The recommended value for inexperienced users is 110 (i.e., hard failure with all messages printed).

Unless the routine detects an error (see Section 6), IFAIL contains 0 on exit.

## 6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

One or more of the parameters N, MNP, NP, NUMBEG, NUMMIX, TOL, DELEPS, LWORK or LIWORK has been incorrectly set, or  $X(1) \geq X(\text{NP})$  or the mesh points  $X(i)$  are not in strictly ascending order.

IFAIL = 2

A finer mesh is required for the accuracy requested; that is MNP is not large enough. This error exit normally occurs when the problem being solved is difficult (for example, there is a boundary layer) and high accuracy is requested. A poor initial choice of mesh points will make this error exit more likely.

IFAIL = 3

The Newton iteration has failed to converge. There are several possible causes for this error:

- (i) faulty coding in one of the Jacobian calculation routines;
- (ii) if IJAC = 0 then inaccurate Jacobians may have been calculated numerically (this is a very unlikely cause); or,
- (iii) a poor initial mesh or initial approximate solution has been selected either by the user or by default or there are not enough points in the initial mesh. Possibly, the user should try the continuation facility.

IFAIL = 4

The Newton iteration has reached round-off error level. It could be however that the answer returned is satisfactory. The error is likely to occur if too high an accuracy is requested.



IFAIL = 5

The Jacobian calculated by JACOBG (or the equivalent matrix calculated by numerical differentiation) is singular. This may occur due to faulty coding of JACOBG or, in some circumstances, to a zero initial choice of approximate solution (such as is chosen when INIT = 0).

IFAIL = 6

There is no dependence on  $\epsilon$  when continuation is being used. This can be due to faulty coding of JACEPS or JACGEP or, in some circumstances, to a zero initial choice of approximate solution (such as is chosen when INIT = 0).

IFAIL = 7

DELEPS is required to be less than *machine precision* for continuation to proceed. It is likely that either the problem (3) has no solution for some value near the current value of  $\epsilon$  (see the advisory print out from D02RAF) or that the problem is so difficult that even with continuation it is unlikely to be solved using this routine. If the latter cause is suspected then using more mesh points initially may help.

IFAIL = 8

IFAIL = 9

Indicates that a serious error has occurred in a call to D02RAF or D02RAR. Check all array subscripts and subroutine parameter lists in calls to D02RAF. Seek expert help.

## 7 Accuracy

The solution returned by the routine will be accurate to the user's tolerance as defined by the relation (5) except in extreme circumstances. The final error estimate over the whole mesh for each component is given in the array ABT. If too many points are specified in the initial mesh, the solution may be more accurate than requested and the error may not be approximately equidistributed.

## 8 Further Comments

There are too many factors present to quantify the timing. The time taken by the routine is negligible only on very simple problems.

The user is strongly recommended to set IFAIL to obtain self-explanatory error messages, and also monitoring information about the course of the computation.

In the case where the user wishes to solve a sequence of similar problems, the use of the final mesh and solution from one case as the initial mesh is strongly recommended for the next.

## 9 Example

We solve the differential equation

$$y''' = -yy'' - 2\epsilon(1 - y^2)$$

with  $\epsilon = 1$  and boundary conditions

$$y(0) = y'(0) = 0, \quad y'(10) = 1$$

to an accuracy specified by TOL = 1.0E-4. The continuation facility is used with the continuation parameter  $\epsilon$  introduced as in the differential equation above and with DELEPS = 0.1 initially. (The continuation facility is not needed for this problem and is used here for illustration.)

### 9.1 Program Text

**Note:** the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```

*      D02RAF Example Program Text
*      Mark 14 Revised.  NAG Copyright 1989.
*      .. Parameters ..
INTEGER      N, MNP, IY, LWORK, LIWORK
PARAMETER   (N=3,MNP=40,IY=N,LWORK=MNP*(3*N*N+6*N+2)
+           +4*N*N+3*N,LIWORK=MNP*(2*N+1)+N)
INTEGER      NOUT
PARAMETER   (NOUT=6)
*      .. Local Scalars ..
real       DELEPS, TOL
INTEGER      I, IFAIL, IJAC, INIT, J, NP, NUMBEG, NUMMIX
*      .. Local Arrays ..
real       ABT(N), WORK(LWORK), X(MNP), Y(IY,MNP)
INTEGER      IWORK(LIWORK)
*      .. External Subroutines ..
EXTERNAL    D02RAF, FCN, G, JACEPS, JACGEP, JACOBG, JACOBG,
+           X04ABF
*      .. Executable Statements ..
WRITE (NOUT,*) 'D02RAF Example Program Results'
WRITE (NOUT,*)
WRITE (NOUT,*) 'Calculation using analytic Jacobians'
CALL X04ABF(1,NOUT)
TOL = 1.0e-4
NP = 17
NUMBEG = 2
NUMMIX = 0
X(1) = 0.0e0
X(NP) = 10.0e0
INIT = 0
DELEPS = 0.1e0
IJAC = 1
*      * Set IFAIL to 111 to obtain monitoring information *
IFAIL = 11
*
CALL D02RAF(N,MNP,NP,NUMBEG,NUMMIX,TOL,INIT,X,Y,N,ABT,FCN,G,IJAC,
+         JACOBG,JACOBG,DELEPS,JACEPS,JACGEP,WORK,LWORK,IWORK,
+         LIWORK,IFAIL)
*
IF (IFAIL.EQ.0 .OR. IFAIL.EQ.4) THEN
  IF (IFAIL.EQ.4) WRITE (NOUT,99996)
+  'On exit from D02RAF IFAIL = ', IFAIL
  WRITE (NOUT,*)
  WRITE (NOUT,99999) 'Solution on final mesh of ', NP, ' points'
  WRITE (NOUT,*)
+  '      X(I)          Y1(I)          Y2(I)          Y3(I)'
  WRITE (NOUT,99998) (X(J),(Y(I,J),I=1,N),J=1,NP)
  WRITE (NOUT,*)
  WRITE (NOUT,*) 'Maximum estimated error by components'
  WRITE (NOUT,99997) (ABT(I),I=1,N)
ELSE
  WRITE (NOUT,99996) 'On exit from D02RAF IFAIL = ', IFAIL
END IF
STOP
*
99999 FORMAT (1X,A,I2,A)
99998 FORMAT (1X,F10.3,3F13.4)
99997 FORMAT (11X,1P,3E13.2)
99996 FORMAT (1X,A,I3)
END
*
SUBROUTINE FCN(X,EPS,Y,F,M)
*      .. Scalar Arguments ..
real       EPS, X
INTEGER      M
*      .. Array Arguments ..
real       F(M), Y(M)
*      .. Executable Statements ..
F(1) = Y(2)
F(2) = Y(3)
F(3) = -Y(1)*Y(3) - 2.0e0*(1.0e0-Y(2)*Y(2))*EPS
RETURN

```

```

      END
*
      SUBROUTINE G(EPS,Y,Z,AL,M)
*      .. Scalar Arguments ..
      real          EPS
      INTEGER      M
*      .. Array Arguments ..
      real          AL(M), Y(M), Z(M)
*      .. Executable Statements ..
      AL(1) = Y(1)
      AL(2) = Y(2)
      AL(3) = Z(2) - 1.0e0
      RETURN
      END
*
      SUBROUTINE JACEPS(X,EPS,Y,F,M)
*      .. Scalar Arguments ..
      real          EPS, X
      INTEGER      M
*      .. Array Arguments ..
      real          F(M), Y(M)
*      .. Executable Statements ..
      F(1) = 0.0e0
      F(2) = 0.0e0
      F(3) = -2.0e0*(1.0e0-Y(2)*Y(2))
      RETURN
      END
*
      SUBROUTINE JACGEP(EPS,Y,Z,AL,M)
*      .. Scalar Arguments ..
      real          EPS
      INTEGER      M
*      .. Array Arguments ..
      real          AL(M), Y(M), Z(M)
*      .. Local Scalars ..
      INTEGER      I
*      .. Executable Statements ..
      DO 20 I = 1, M
          AL(I) = 0.0e0
20  CONTINUE
      RETURN
      END
*
      SUBROUTINE JACOBF(X,EPS,Y,F,M)
*      .. Scalar Arguments ..
      real          EPS, X
      INTEGER      M
*      .. Array Arguments ..
      real          F(M,M), Y(M)
*      .. Local Scalars ..
      INTEGER      I, J
*      .. Executable Statements ..
      DO 40 I = 1, M
          DO 20 J = 1, M
              F(I,J) = 0.0e0
20  CONTINUE
40  CONTINUE
      F(1,2) = 1.0e0
      F(2,3) = 1.0e0
      F(3,1) = -Y(3)
      F(3,2) = 4.0e0*Y(2)*EPS
      F(3,3) = -Y(1)
      RETURN
      END
*
      SUBROUTINE JACOBG(EPS,Y,Z,A,B,M)
*      .. Scalar Arguments ..
      real          EPS
      INTEGER      M
*      .. Array Arguments ..
      real          A(M,M), B(M,M), Y(M), Z(M)

```

```

*    .. Local Scalars ..
      INTEGER          I, J
*    .. Executable Statements ..
      DO 40 I = 1, M
        DO 20 J = 1, M
          A(I,J) = 0.0e0
          B(I,J) = 0.0e0
        20 CONTINUE
      40 CONTINUE
      A(1,1) = 1.0e0
      A(2,2) = 1.0e0
      B(3,2) = 1.0e0
      RETURN
      END

```

## 9.2 Program Data

None.

## 9.3 Program Results

D02RAF Example Program Results

Calculation using analytic Jacobians

Solution on final mesh of 33 points

X(I)	Y1(I)	Y2(I)	Y3(I)
0.000	0.0000	0.0000	1.6872
0.062	0.0032	0.1016	1.5626
0.125	0.0125	0.1954	1.4398
0.188	0.0275	0.2816	1.3203
0.250	0.0476	0.3605	1.2054
0.375	0.1015	0.4976	0.9924
0.500	0.1709	0.6097	0.8048
0.625	0.2530	0.6999	0.6438
0.703	0.3095	0.7467	0.5563
0.781	0.3695	0.7871	0.4784
0.938	0.4978	0.8513	0.3490
1.094	0.6346	0.8977	0.2502
1.250	0.7776	0.9308	0.1763
1.458	0.9748	0.9598	0.1077
1.667	1.1768	0.9773	0.0639
1.875	1.3815	0.9876	0.0367
2.031	1.5362	0.9922	0.0238
2.188	1.6915	0.9952	0.0151
2.500	2.0031	0.9983	0.0058
2.656	2.1591	0.9990	0.0035
2.812	2.3153	0.9994	0.0021
3.125	2.6277	0.9998	0.0007
3.750	3.2526	1.0000	0.0001
4.375	3.8776	1.0000	0.0000
5.000	4.5026	1.0000	0.0000
5.625	5.1276	1.0000	-0.0000
6.250	5.7526	1.0000	0.0000
6.875	6.3776	1.0000	-0.0000
7.500	7.0026	1.0000	0.0000
8.125	7.6276	1.0000	-0.0000
8.750	8.2526	1.0000	0.0000
9.375	8.8776	1.0000	-0.0000
10.000	9.5026	1.0000	0.0000

Maximum estimated error by components

6.92E-05	1.81E-05	6.42E-05
----------	----------	----------